

    
      
          
            
  
We moved to Github!

This documentation is outdated. The official documentation was moved to
Formality’s repository [https://github.com/moonad/Formality/blob/master/DOCUMENTATION.md].

[image: _images/giphy.gif]

Why Formality-Core?

Formality-Core (FMC) is a minimal, efficient functional programming language designed to be a “low-level” compile-target for more feature-rich functional languages. Differently from most functional languages, FMC has amazing computational characteristics such as not requiring a garbage-collector, beta-reduction being a constant-time operation, being able to perform dynamic fusion, being strongly normalizing and massively parallelizable. In other words, FMC is really, really fast, and aims to be the ultimate assembly of resource-aware functional languages.

In order for all that to be possible, FMC makes some fundamental changes to the memory model behind the λ-calculus, allowing it to be compatible with an optimal functional runtime as Lamping’s Abstract Algorithm [https://medium.com/@maiavictor/solving-the-mystery-behind-abstract-algorithms-magical-optimizations-144225164b07], which is based on a beautiful model of computation, Interaction-Nets [https://pdfs.semanticscholar.org/1731/a6e49c6c2afda3e72256ba0afb34957377d3.pdf]. The most important change is the fact its lambdas are affine: a lambda-bound variables can’t be used more than once. At first, this sounds like an extreme limitation, but this is mitigated by FMC’s boxed duplication system. It allows you to perform deep copy of arbitrary values, as long as you respect certain structural restrictions.

Formality-Core is very similar to Rust. The main difference is that Rust mitigates the limitations of affine lambdas with a complex ownership system which includes borrows, lifetimes, mutable references. In FMC, you’re instead encouraged to simply .clone() structures whenever you need to. Due to the underlying interaction-net runtime, those copies are performed lazily, granularly and in parallel. They’re, thus, much less expensive than a Rust-like .clone() (and often free).

Here is an brief illustration of the process of compiling a Formality-Core term to our interaction-net runtime, reducing it in parallel, and reading the result:

[image: https://github.com/moonad/formality/raw/master/docs/images/inet-simulation.gif]





          

      

      

    

  

    
      
          
            
  
What are proofs?

Since proofs are, for most devs, the most unusual aspect of Formality, I’ve put this section on the docs as a simple explanation on what they are useful for. I think the best way to describe it is by using the analogy of “specifications as types”. Let me elaborate. There is a full spectrum of type systems, right? JS and Python are untyped: the input of a function can be anything, nothing is guaranteed. In C, Java, Solidity, you can be more precise: “this function accepts an int and returns an int”. That “specifies” what the function does to an extent, but is very limited. Formality is at the top of the ladder: its types are so precise that you can specify a complete algorithm at the type language! Let me give you an example:

import Base@0

Spec : Type
  {a : Bool} -> [b : Bool ~ Not(Equal(Bool, %a, %b))]





This code specifies “a function that receives a Bool a, and returns a bool b, such that a != b”. Can you see how there is only one Bool -> Bool function that satisfies that specification? And the cool thing is that the compiler can verify if a function satisfies it mechanically, without room for error. So, this works:

// A function that negates a boolean
negate : {case a : Bool} -> [b : Bool ~ Not(a == b)]
| true  => [false ~ true_not_false] // if a is true, return false, and prove that `a != false`
| false => [true  ~ false_not_true] // if a is false, return true, and prove that `a != true"

// Proves that the "negate" function satisfies our `Spec`
main : Spec
  negate





But if you change any of the returned bools, it won’t work anymore. It is literally impossible to make anything other than a boolean negation pass! In other words, a term of type Spec proves the specification represented by it. That’s what theorem proving is; it is nothing but a fancy way to say “type-checking in a language that has very precise types”. And Formality types can be arbitrarily precise. For example, this Spec:

Spec : Type
  {len : Ind} ->                       // Given a length `len`
  ! { ~A : Type                        // And a type `A`
    , idx : Fin(len)                   // And an index, `idx`, up to `len`
    , vec : Vec(A, len)                // And a vector, `vec`, with `len` elements of type `A`
    } -> [x : A ~ At(A,x,len,idx,vec)] // Returns the element `x` that is at index `idx` of that `vec`





specifies an array accessor that can’t have an out-of-bounds error, and that can’t return the wrong element! If OpenSSL proved it, we wouldn’t have Heartbleed. And the cool thing is that those proofs happen statically, they have zero runtime costs!

In the context of smart-contracts, we could have specs like “this contract’s balance satisfies certain invariant”, completely preventing things like TheDAO being drained. Of course, proofs can be huge and ugly, but that’s ok, developers are paid to work hard and write good software. The point is to have a small list of simple specifications that users can read and be confident the smart-contract behaves as desired, without having to trust its developers.

From this Reddit thread [https://www.reddit.com/r/ethereum/comments/d45vpq/im_hyper_bullish_on_ethereum/f08waxj/?context=1].





          

      

      

    

  

    
      
          
            
  
Why not X instead?

There are amazing projects with similar goals and vision. In this article, I’ll point the reasons we’ve decided to build Formality rather than using one of them.


Why not Plutus, Michelson or Vyper instead?

Okay, so, Plutus in particular is just a “normal” functional language, very similar to Haskell. It has algebraic datatypes and polymorphic functions, which is actually really cool if compared to Solidity, but that’s about it. It has none of the tools required to qualify as a “proof language”.

No dependent types.

You can’t have values at the type level at all, so, while you can have polymorphic types like “a List of numbers”, you can’t have really precise types like “a List with 8 even numbers”.

No induction or inductive datatypes.

As any mathematician may tell you, you can’t go much far in mathematics without induction. Even proving something as simple as a + b = b + a requires it. Imagine properties about smart-contracts.

No equality types.

Even if you could live without induction, you can’t even express something like a + b = b + a on Plutus. There is no concept of equality on its type system.

No sigmas (“such that”).

Plutus has a polymorphic “forall”, but no “there exists” or “such that”. So, you can’t say something like “give me a function that returns a number such that it is a multiple of 7…”.

And so on.

In short, Plutus is just a normal, non-proof language, it simply doesn’t have a type language capable of expressing specifications and theorems in any way. The only sense in which I could see someone selling Plutus as more proof-friendly than Solidity would be that it is can be imported in a “real” proof language like Agda. But the same is true for the EVM, and even if it wasn’t, it is awfully inconvenient. Formality can perfectly prove theorems about its own programs.

TODO: elaborate about Michelson and Solidity (generally, the same points apply).

From this Reddit thread [https://www.reddit.com/r/ethereum/comments/d45vpq/im_hyper_bullish_on_ethereum/f09cj2f/?context=2].




Why not Agda, Coq or Idris instead?

TODO: write this section including points such as:

Runtime efficiency and portabiliy.

Write about the difficulty of compiling non-linear languages to inets efficienly. Full λ-calculus closures require high-order, garbage-collected runtimes that are generally too expensive for blockchains and similar. Also write about Formality’s 450 LOC runtime and how that matters for portability.

Implementation complexity.

Write about our experience trying to hack and extend Agda/Idris with features we needed.

Compiler portability.

Write about the difficulty of running Agda/Idris inside a browser or mobile efficiently, for applications such as Provit, or a web-based smart-contract verifying site, etc.

Logical consistency.

We can’t fix potential bugs in other languages and we absolutely require consistency for our applications, so a small stable core is important. Agda doesn’t have one, Idris has inconsistency bugs. I think Coq would be fine though.

(Note about how much we love Agda and Idris and are actually constantly trying to use those instead, and re-evaluating whether it’d be possible.)







          

      

      

    

  

    
      
          
            
  
Motivation

Formality exists to fill a hole in the current market: there aren’t many languages featuring theorem proving that are simple, user-friendly and efficient. To accomplish that goal, we rely on several design philosophies:


An accessible syntax

Proof languages often have complex syntaxes that make them needlessly inaccessible, as if the subject wasn’t hard enough already. Coq, for example, uses 3 different languages with different rules and an overall heavy syntax. Agda is clean and beautiful, but relies heavily on unicode and agda-mode, making it essentially unusable outside of EMACs, which is arguably a “hardcore” editor. Formality aims to keep a simple, familiar syntax that is much closer to common languages like Python and JavaScript. A regular TypeScript developer should, for example, be able to read our Functor [https://github.com/moonad/Formality-Base/blob/master/Control.Functor.fm] formalization without extensive training. While we may not be quite there, we’re making fast progress towards that goal.




Fast and portable “by design”

Some languages are inherently slow, by design. JavaScript, for example, is slower than C: all things equal, its mandatory garbage collector will be an unavoidable disadvantage. Formality is meant to be as fast as theoretically possible. For example, it has affine lambdas, allowing it to be garbage-collection-free. It has a strongly confluent interaction-net runtime, allowing it to be evaluated in massively parallel architectures. It doesn’t require bruijn bookkeeping, making it the fastest “closure chunker” around. It is lazy, it has a clear cost model for blockchains, it has a minuscle (448 LOC [https://github.com/moonad/Formality/blob/master/src/fm-net.js] runtime that can easily be ported to multiple platforms. Right now, Formality’s compiler isn’t as mature as the ones found in decades-old languages, but it has endless room for improvements, since the language is fast “by design”.




An optimal high-order evaluator

Formality’s substitution algorithm is asymptotically faster than Haskell’s, Clojure’s, JavaScript’s and other closure implementations. This makes it extremely fast at evaluating high-order programs, combining a Haskell-like high-level feel with a Rust-like low-level performance curve. For example, Haskell’s stream fusion, a hard-coded, important optimization, happens naturally, at runtime [https://medium.com/@maiavictor/solving-the-mystery-behind-abstract-algorithms-magical-optimizations-144225164b07], on Formality. This also allow us to explore new ways to develop algorithms, such as this “impossibly efficient” exp-mod implementation [https://medium.com/@maiavictor/calling-a-function-a-googol-times-53933c072e3a] implementation. Who knows if this may lead to new breakthroughs in complexity theory?




An elegant underlying Type Theory

Formality’s unique approach to termination is conjectured to allow it to have elegant, powerful type-level features that would be otherwise impossible without causing logical inconsistencies. For example, instead of built-in datatypes, we rely on Self Types [https://www.semanticscholar.org/paper/Self-Types-for-Dependently-Typed-Lambda-Encodings-Fu-Stump/652f673e13b889e0fd7adbd480c2fdf290621f66], which allow us to implement inductive families with native lambdas. As history tells, having elegant foundations often pays back. We’ve not only managed to port several proofs from other assistants, but found techniques to emulate Coq’s structural recursion [https://github.com/moonad/Formality-Base/commit/b777d806c6fa37f2ce306fbe87b3ed267152b90c], to perform large eliminations, and even an hypothetical encoding of higher inductive types [https://github.com/moonad/Formality-Base/blob/master/Example.HigherInductiveType.fm]; and we’ve barely began exploring the system.

[image: ../_images/inet-simulation.gif]

Interaction Net (inet) simulation







          

      

      

    

  

    
      
          
            
  
Installation

Right now, Formality can only be installed through npm. Install npm following
this guide [https://www.npmjs.com/get-npm]. Then, go to the command-line and
type:

$ npm i -g formality-lang





Note that Formality requires Node.js v0.12:

$ node -v
v12.5.0





Using node2nix [https://github.com/svanderburg/node2nix#installation], we can also install Formality using the Nix package manager:

$ git clone git@github.com:moonad/Formality.git
$ cd Formality
$ nix-channel --add https://nixos.org/channels/nixpkgs-unstable unstable
$ nix-env -f '<unstable>' -iA nodePackages.node2nix
$ node2nix --nodejs-12
$ sed -i 's/nixpkgs/unstable/g' default.nix
$ nix-env -f default.nix -iA package





This should be all you need. In order to test if it worked, type fm on the
terminal. If you see Formality’s command-line options, then it has been
successfully installed in your system. If you have any problem during this
process, please open an issue [https://github.com/moonad/Formality/issues].





          

      

      

    

  

    
      
          
            
  
Hello, World!

This is the “Hello, World!” in Formality:

import Base@0

main : Output
  print("Hello, world!")





Save this file as hello.fm.


Running (debug)

To run it, type fm hello/main. This will evaluate main using an interpreter in debug mode and output "Hello, world!".




Running (fast)

To run it with the interaction-net runtime, type fm -o hello/main. This will be faster, but you’ll lose information like variable names and logs.




Type-checking

To type-check it, type fm -t hello/main. This will check if the program’s type is correct and print it (in this case, Output). If the type is incorrect, it will print an error message. For example, if you change "Hello, world!" to 7, it will print:

Type mismatch.
- Found type... Number
- Instead of... String
- When checking 7
- On line 4, col 9, file _.fm:
  1| import Base@0
  2|
  3| main : Output
  4|   print(7)
  5|





Because 7 is a Number, but the print function expects a String.







          

      

      

    

  

    
      
          
            
  
Core Features


Let

Allows you to give local names to terms.

import Base@0

main : Output
  let hello = "Hello, world!"
  print(hello)





let expressions can be infinitely nested.

import Base@0

main : Output
  let output =
    let hello = "Hello, world!"
    print(hello)
  output





Note: let has no computational effect, it simply performs a parse-time substitution.




Numbers

The type of a native number is Number.

main : Number
  1900





They can also be written in hexadecimal:

main : Number
  0x76C





And in binary:

main : Number
  0b11101101100





They include many built-in operations:




	name
	syntax
	javascript equivalent





	addition
	x .+. y
	x + y



	subtraction
	x .-. y
	x - y



	multiplication
	x .*. y
	x * y



	division
	x ./. y
	x / y



	modulus
	x .%. y
	x % y



	exponentiation
	x .**. y
	x ** y



	bitwise-and
	x .&. y
	x & y



	bitwise-or
	x .|. y
	x | y



	bitwise-xor
	x .^. y
	x ^ y



	bitwise-not
	.~.(y)
	~y



	bitwise-right-shift
	x .>>>. y
	x >>> y



	bitwise-left-shift
	x .<<. y
	x << y



	greater-than
	x .>. y
	x > y ? 1 : 0



	less-than
	x .<. y
	x < y ? 1 : 0



	equals
	x .==. y
	x === y ? 1 : 0




There is no operator precedence: parenthesis are always placed on the right. That means 3 * 10 + 1 is parsed as 3 * (10 + 1). If you want the multiplication to occur first, you must be explicit:

main : Number
  (3 .*. 10) .+. 1








If / Else

if allows branching with a Number condition.




	syntax
	description





	if n: a else: b
	If n .= 0, evaluates to b, else, evaluates to a




Usage is straightforward:

import Base@0

main : Output
  let age = 30

  if age .<. 18:
    print("boring teenager")
  else:
    print("respect your elders!")








Pairs

Native pairs store two elements of possibly different types.




	syntax
	description





	[x : A, B(x)]
	The type of a pair



	[a, b]
	Creates a pair with elements a and b



	fst(p)
	Extracts the first element of a pair



	snd(p)
	Extracts the second element of a pair



	get [a, b] = p ...
	Extracts both elements of a pair




Note that the type of a pair is [x : A, B(x)], because the type of the second element can depend on the value of the first. When it doesn’t, you can write just [:A, B] instead. Using pairs is straightforward. Examples:

Creating:

main : [:Number, Number]
  [1, 2]





Extracting the first element:

main : Number
  let pair = [1, 2]
  fst(pair)





Extracting both elements:

main : Number
  let pair  = [1, 2]
  get [a,b] = pair
  a .+. b





Nesting to the left:

import Base@0

main : [:[:Number, Number], String]
  [[1, 2], "Hello World!"]





Nesting to the right:

main : Number
  let triple  = [1, 2, 3] // same as [1, [2, 3]]
  get [x,y,z] = triple
  x .+. y .+. z





Erased (first element):

main : [~: Number, Number]
  [~1, 2] // the number "1" is erased from runtime





Erased (second element):

main : [: Number ~ Number]
  [1 ~ 2] // the number "2" is erased from runtime





Notably, the second element of a pair can depend on the value of the first.

main : [x : Number, (if x: Number else: Bool)]
  [0, true] // if you change 0 to 1, the second element must be a Number.








Functions

Formality functions are anonymous expressions, like Haskell’s lambdas. There are no multi-argument lambdas.




	syntax
	description





	{x : A, y : B, z : C, ...} -> D
	Function type with args x : A, y : B, z : C, returning D



	{x, y, z, ...} body
	A function that receives the arguments x, y, z and returns body



	f(x, y, z, ...)
	Applies the function f to the arguments x, y, z (curried)




Formality functions are anonymous expressions, like Haskell’s lambdas. There are no multi-argument lambdas; {x, y, z, ...} body is the same as {x} {y} {z} ... body, which works like JS’s x => y => z => ... body and Haskell’s \ x y z ... -> body. Function calls use f(x, y, z) syntax, which is the same as f(x)(y)(z).... The type of a function is written as A -> B -> C -> D, like on Haskell, but it can also be written with names, as {x : A, y : B, z : C ...} -> D, which is equivalent to Agda’s (x : A) -> (y : B) -> (z : C) -> ... D. Examples:

A top-level function:

adder : Number -> Number -> Number
  {x, y} x .+. y

main : Number
  adder(40, 2)





When you write the variable names, lambdas are added implicity. For example:

adder : {x : Number, y : Number} -> Number
  x .+. y

main : Number
  adder(40, 2)





An inline function (lambda expression):

main : Number
  ({x : Number, y : Number} x + y)(40, 2)





You can annotate the full type rather than the type of each variable:

main : Number
  (({x, y} x .+. y) :: Number -> Number -> Number)(40, 2)





You can avoid types. This won’t type-check, but can still be ran:

main
  ({x, y} x .+. y)(40, 2)





Lambdas and applications can be erased with a ~, which causes them to vanish from the compiled output. This is useful, for example, to write polymorphic functions without extra runtime costs. For example, on the code below, id is compiled to {x} x, and main is compiled to id(42). The first argument disappears from the runtime.

id : {~T : Type, x : T} -> T
  x

main : Number
  id(~Number, 42)





Formality functions are affine, which means you can’t use a variable more than once. For example, the program below isn’t allowed, because b is used twice:

copy : {b : Bool} -> [:Bool, Bool]
  [b, b]





While this sounds limiting, there are many ways to deal with it, as will be explained later, and it is extremelly important for both logical consistency and runtime performance.




Boxes and Copying

Formality includes primives for performing explicit, deep copies of terms, as long as they’re “boxed”.




	syntax
	description





	#t
	Puts term t inside a box



	!T
	The type of a boxed term



	dup x = t; u
	Unboxes t and copies it as x inside u



	$t
	Unboxes t




Since this increases the power of the language considerably, in order not to cause logical inconsistencies, and to still allow for an efficient runtime, boxes are limited by the “stratification condition”. It enforces that the number of #s surrounding a term must never change during reduction. As such, boxes aren’t very useful for copying data, but are essential to implement control structures like loops and recursion. This all will be explained in more details later on.




Equality

Formality used to include untyped equality primitives, the same one seen in Cedille [https://github.com/cedille/cedille]. It doesn’t anymore because most of its use cases can be replicated with a user-defined equality type [https://github.com/moonad/Formality-Base/blob/master/Data.Equal.fm]. This will be explained in an advanced tutorial. To learn about the old equality, click here [https://gist.github.com/MaiaVictor/a2cdceb2041ebec56cab5de643f1a761], as it can still be pedagogical.




Self Types

Formality also has Self Types [http://homepage.divms.uiowa.edu/~astump/papers/fu-stump-rta-tlca-14.pdf], which allow it us to implement inductive datatypes with λ-encodings:




	syntax
	description





	${self} T(self)
	T is a type that can access its own value



	new(~T) t
	Constructs an instance of a T with value t



	(%t)
	Consumes a self-type t, giving its type access to its value




Note that Self Types are not the same as recursive types. Recursive types allow the type to access itself. For example:

N_Nums : {n : Number} -> Type
  if n .<. 2:
    Number
  else:
    [:Number, N_Nums(n .-. 1)]





The type above allows you to create a list of N words:

main : N_Nums(4)
  [0, 1, 2, 3]





Self Types allow a type to access its own value. This has many uses. Suppose that you wanted to create a pair of identical words:

sameNums : Type
  ${self} [:Number, :Number, fst(self) == fst(snd(self))]

same_words_0 : sameNums 
  new(~sameNums) [0, 0, res]

same_words_1 : sameNums 
  new(~sameNums) [1, 1, res]

same_words_2 : sameNums 
  new(~sameNums) [2, 2, refl(~2)]





Notice how the sameNums type has access to values of its terms. So, when we instantiate same_words_0, the self variable on the type is replaced by [0, 0, refl(~0)], which becomes [:Num, :Num, 0 == 0], allowing us to write the last element a refl(~0). Of course, in this case, this effect could be achieved with dependent pairs:

sameNumsB : Type
  [x : Number, y : Number, Equal(Number, %x, %y)]

same_words_b_0 : sameNumsB
  [0, 0, refl(~Number, ~%0)]

same_words_b_1 : sameNumsB
  [1, 1, refl(~Number, ~%1)]

same_words_b_2 : sameNumsB
  [2, 2, refl(~Number, ~%2)





But what is interesting is that Self Types are more expressive than dependent pairs. We could, for example, make an “insanely dependent” pair where the first element depends on the second, and vice-versa:

insane : Type
  ${self}
  [: [a : Number, (a .*. 2) == fst(snd(self))],
     [b : Number, b == fst(fst(self)) .+. 1]]
  
insane : insane
  new(~insane)
  [[1, refl(~2)],
   [2, refl(~2)]]





This define the type of pairs [a,b] such that a .*. 2 = b and b = a .+. 2. Notice the mutual reference. The only possible way to make it is with a = 1 and b = 2, which is the solution to the equation above.

Self Types are used by Formality to create inductive datatypes. This is explained in more details on the tutorial section.




Annotations

You can also explictly annotate the type of a term:




	syntax
	description





	term :: Type
	Annotates term with type Type




This is useful when the bidirectional type-checker can’t infer the type of an expression.

main : Number
  (({x, y} x .+. y) :: Number -> Number -> Number)(40, 2)





They’re also important for dependent pairs:

annotation0
   [1, 2]

annotation1
   [1, 2] :: [x : Number, (if x: Number else: Bool)]





The pairs above have different types, despite having the same value. Explicit annotations are inline, different from top-level annotations:

annotation2 : [x : Number, (if x: Number else: Bool)]








Non-terminating terms

While Formality’s stratification checks make it a terminating language with a well-defined complexity class, this is relaxed in certain parts of your programs: inside types and unrestricted terms. Those bring first-class non-termination to the language, and can be expressed as:




	syntax
	description





	-A
	A is an unrestricted term of type A



	%t
	Converts t to an unrestricted term



	+t
	Converts t back to a restricted term




In general, prefixing a type with - means that it can either be a well-typed term of that type or non-terminating computation. The %t syntax marks a term as unrestricted, and the +t syntax removes the mark, but it can only be used inside other unrestricted terms, or types. Inside both types and unrestricted terms, you can unbox terms at will, use variables more than once, and even use top-level definitions recursively. So, for example, while the List(Number) type represents a finite list of words, the -List(Number) type can be infinite:

  // A list with infinite copies of the number 1
  ones : -List(Number)
    % cons(~Number, 1, +ones)





And you also use unrestricted terms to write recursive algorithms:

  double : {x : Number} -> -Number
    if x .==. 0:
      % 0
    else:
      % 2 .+. +double(x .-. 1)





The trade-off is that:


	Unrestricted terms can’t be compiled to efficient interaction nets.


	Unrestricted terms can’t be interpreted as valid mathematical proofs.




The reason for (1) is that, without stratification checks, Formality terms stop being compatible with our efficient, “oracle-free” interaction net runtime. The reason for (2) is that, with unrestricted recursion, it is easy to prove absurd statements. For example, one could prove -Empty by just making a loop:

  main : -Empty
    %+main





But this can’t be taken as a mathematical proof. Note that normal Formality terms (not marked with a -) can be seen as mathematical proofs, even if their types refer to non-terminating terms (ex: double(x) == add(x, x) for recursively defined double and add).  That’s because our core language is the Elementary Affine Lambda Calculus, which is not only very restricted computationally, but terminates independent of types.

Unrestricted terms are extremely useful when you have a function argument that will only be used inside a type; for example, the Nat index of a Vector. They also play an important role in the implementation of inductive datatypes with self-types.




Holes

Formality also features holes, which are very useful for development and debugging. A hole can be used to fill a part of your program that you don’t want to implement yet. It can be written anywhere as ?name, with the name being optional. If you give it a name, it will cause Formality to print the type expected on the hole location, its context (scope variables), and possibly a value, if Formality can fill it for you. For example, the program below:

import Base@0

main : {x : Bool} -> Bool
  and(true, ?a)





Will output:

[ERROR]
Found hole: 'a'.
- With goal... Bool
- Couldn't solve it.
- With context:
- x : Bool





This tells you that, on the location of the hole, you should have a Bool. In some cases, Formality will be able to find a well-typed term that could be used to fill this hole. For example, on the code above:

main : List(Number)
  cons(~?a, 2, nil(~Number))





Formality outputs:

Found hole: 'a'.
- With goal... Type
- Solved as... Number





Since Formality can infer that the value of ?a should be Number. You can proceed to fill it, or just remove the name, leaving only a ?: this will cause the program to type-check, since Formality knows how to complete it. As a shorthand, you can use _ instead of ~?:

main : List(Number)
  cons(_2, nil(_))





Even when Formality can’t fill a hole, it will locally assume it to be true, allowing you to move on to other parts of your program before returning, making them very useful for development. Note that this is only automatic if Formality can infer the expected type of the hole’s location. Otherwise, you must give it an explicit annotation, as in ?hole :: MyType. Of course, unfilled holes cause the top-level term to fail to type-check.




Logs

Another handy feature is log(x). When running a program, it will print the normal form of x, similarly to haskell’s console.log and haskell’s print, but for anything (not only strings). When type-checking a program, it tells you the normal-form and the type of x. This is useful when you want to know what type an expression would have inside certain context. For example:

import Base@0

main : {f : Bool -> Nat} -> Nat
  log(f(true))
  ?a





Type-checking the program above will cause Formality to output:

[LOG]
Term: f(true)
Type: Nat

Found hole: 'a'.
- With goal... Nat
- Couldn't solve it.
- With context:
- f : {:Bool} -> Nat

Unsolved holes.





This tells you that, inside the body of main, the type of f(true) is Nat. Since it coincides with the type of the hole, you can complete the program above with it:

import Base@0

main : {f : Bool -> Nat} -> Nat
  f(true)





Compile-time logs are extremelly useful for development. We highly recommend you to use them as much as possible!




Local imports

The import statement can be used to include local files. For example, save an Answers.fm file in the same directory as hello.fm, with the following contents:

import Base@0

everything : String
  "42"





Then save a test.fm file as:

import Base@0
import Answers

main : Output
  print(everything)





And run it with fm test/main. You should see 42.

If multiple imports have conflicting names, you can disambiguate with File/name, or with a qualified import, using as:

import Base@0
import Answers as A

main : Output
  print(A/everything)








Global imports

Formality also has a file-based package manager. You can use it to share files with other people. A file can be saved globally with fm -s file. This will give it a unique name with a version, such as file@7. Once given a unique name, the file contents will never change, so file@7 will always refer to that exact file. As soon as it is saved globally, you can import it from any other computer. For example, remove Answers.fm and change hello.fm to:

import Base@0
import Answers@0

main : Output
  print(everything)





This will load Answers@0.fm inside the fm_modules directory and load it. Any import ending with @N refers to a unique, immutable, permanent global file. That prevents the infamous “dependency hell”, and is useful for many applications.

Right now, global imports are uploaded to our servers, but, in the future, they’ll upload files to decentralized storage such as IPFS/Swarm, and give it a unique name using Ethereum’s naming system.







          

      

      

    

  

    
      
          
            
  
Datatypes

Formality includes a powerful datatype system. A new datatype can be defined with the T syntax, which is similar to Haskell’s data, and creates global definitions for its type and constructors. To pattern-match against a value of a datatype, you must use case/T.


Enumerations

Enumerations can be defined and used as follows:

import Base@0

T Suit
| clubs
| diamonds
| hearts
| spades

print_suit : {suit : Suit} -> Output
  case/Suit suit
  | clubs    => print("First rule: you do not talk about Fight Club.")
  | diamonds => print("Queen shines more than diamond.")
  | hearts   => print("You always had mine.")
  | spades   => print("The only card I need is the Ace of Spades! \m/")
  : Output

main : Output
  print_suit(spades)





The program above creates a datatype, Suit, with 4 possible values. In Formality, we call those values constructors. It then pattern-matches a suit and outputs a different sentence depending on it. Notice that the case expression requires you to annotate the returned type: that’s called the motive, and is very useful when theorem proving. You can omit it by using a case argument instead:

print_suit : {case suit : Suit} -> Output
| clubs    => print("First rule: you do not talk about Fight Club.")
| diamonds => print("Queen shines more than diamond.")
| hearts   => print("You always had mine.")
| spades   => print("The only card I need is the Ace of Spades! \m/")





Differently from case expressions, case arguments are shorter and simpler, but are less flexible.




Constructor with fields

Datatype constructors can have fields, allowing them to store values:

import Base@0

T Person
| person {age : Num, name : String}

get_name : {p : Person} -> String
  case/Person p
  | person => p.name
  : String

main : Output
  let john = person(26, "John")

  print(get_name(john))





As you can see, fields can be accessed inside case expressions and case arguments. Notice that p.name is not a field accessor, but just a single variable: the . is part of its name. When Formality doesn’t know the name of the matched value, you must must explicitly name it using the as keyword:

main : {p : Person} -> Num
case/Person person(26, "John") as john
| person => john.age
: Num








Moving resources to branches

Since Formality functions are affine, you can’t use an argument more than once. So, for example, the function below isn’t allowed:

import Base@0

main : {a : Bool, b : Bool} -> Bool
  case/Bool a
  | true  => b
  | false => not(b)
  : Bool





In this case in particular, you used b in two different branches, so you shouldn’t need to copy it:

import Base@0

  main : {a : Bool, b : Bool} -> Bool
  case/Bool a
  + b : Bool
  | true  => b
  | false => not(b)
  : Bool





Under the hoods, this is desugared to an extra lambda on each branch:

import Base@0

main : {a : Bool, b : Bool} -> Bool
  (case/Bool a
  | true  => {b} b
  | false => {b} not(b)
  : Bool -> Bool)(b)





Using case arguments:

import Base@0

main : {case a : Bool, b : Bool} -> Bool
| true  => b
| false => not(b)





As you can see, the version using case arguments is the shortest.




Dependent motives

In Formality, the type returned by a case expression can depend on the matched value. For example, we can do this:

import Base@0

main : Num
  case/Bool true as x
  | true  => 42
  | false => "hello"
  : case/Bool x
    | true  => Num
    | false => String
    : Type





While strange-looking, this is perfectly logical and well-typed. The reason this works is that, when checking the type of a case expression, Formality first specializes the motive for every possible branch to determine “what it demands”. In this case, it demands a Num on the true branch and a String on the false branch. If you satisfy every demand, then it determines the type of the whole case expression by specializing the motive using actual matched value. In this case, we matched on true, so it returns Num.

This has many interesting effects and applications.


Functions with different return types

We could write a function that returns different types based on its input:

import Base@0

foo : {case b : Bool} -> case/Bool b | true => Num | false => String : Type
| true  => 42
| false => "hello"

main : Num
  foo(true) .+. foo(true)





Note this is not the same as returning Either, which is done in Haskell when we want to return different types, needing an extra pattern-match every time you call the function. Here, that’s not necessary, because foo really returns different types based on its input. Thanks to dependent types, Formality can statically determine if you’re doing the right thing, so, replacing foo(true) by foo(false) would be a type error.




Proving absurds with Empty

Another interesting example comes from the Empty datatype:

// {- From Empty@0 -}

T Empty





This code isn’t incomplete, the datatype has zero constructors. This means it is impossible to construct a term t with type Empty, but we can still accept it as a function argument. So, what happens if we pattern match against it?

import Base@0

wtf : {x : Empty} -> ???
  case/Empty
  : ???





The answer is: we can replace ??? by anything, and the program will check. That’s because we technically proved all the demanded cases, so Formality just returns the motive directly as the type of this case expression. That allows us to write a function that returns something absurd:

import Base@0

prove_1_is_2 : {e : Empty} -> Equal(Num, %1, %2)
  case/Empty e
  : Equal(Num, %1, %2)





If we managed to call it, we’d have a proof that 1 .==. 2, which is absurd, making Formality inconsistent. But this is fine: prove_1_is_2 can’t ever be called, because we can’t construct a value of type Empty. This is pretty useful, so we have a base-lib function that, given an element of Empty, returns any absurd type:

// {- From Empty@0 -}

absurd : {e : Empty, ~P : Type} -> P
  case/Empty e
  : P





The opposite holds too: given an absurd equality, we can make an element of type Empty. Here is an example:

// {- From Bool@0 -}

true_isnt_false : {e : true == false} -> Empty
  unit :: rewrite x
    in (case/Bool x
       | true  => Unit
       | false => Empty
       : Type)
    with e





As an exercise, convince yourself why this works.




Unreachable branches with equality notes

Notice the program below:

import Base@0

main : Num
  case/Bool true
  | true  => 10
  | false => ?
  : Num





Here, we’re matching against true, so we know the false case is unreachable, but we still need to fill it with something. In this case, we could write any number, but that’s not always possible. In those cases, equality notes can be helpful:

import Base@0

main : Num
  case/Bool true as x
  + refl(~Bool, ~%true) as e : Equal(Bool, %x, %true)
  | true  => 10
  | false => absurd(false_isnt_true(e), ~Num)
  : Num





This moves a proof that Equal(Nat, x, true) that will be specialized to the value of x on each branch. On the false branch, we get an e : Equal(Nat, false, true), which is absurd, so we can fill it by calling absurd function on e, without providing an actual number.

To desugar an equality note, Formality simply adds an extra, erased equality argument:

import Base@0

main
  (case/Bool true as x
  | true  => {e} 10
  | false => {e} absurd(false_isnt_true(e), ~Num)
  : {e : Equal(Bool, %x, %true)} -> Num)(refl(~Bool, ~%true))





Notice that the left side of the equality note is allowed to access the matched value (x), while the right side is used to build the refl proof.






Recursive fields

Fields can refer to the datatype being defined:

// {- From Nat@0 -}
T Nat
| succ {pred : Nat}
| zero





Since Nat is so common, there is a syntax-sugar for it: 0n3, which expands to succ(succ(succ(zero))).

Mutual recursion is allowed:

T Foo
| foo {bar : Foo}

T Bar
| bar {foo : Bar}





And negative occurrences too (soundly [https://www.reddit.com/r/haskell/comments/d2gcyw/just_letting_you_know_that_formality_has_evolved/ezvjz5m/]):

T Loop
| loop {f : Foo -> Foo}





While recursion is very liberal on types, the same isn’t true for programs. You can’t write recursive functions directly, such as:

mul_by_2 : {case n : Nat} -> Nat
| succ => succ(succ(mul_by_2(n)))
| zero => zero

main : Nat
  mul_by_2(0n2)





This makes dealing with recursive datatypes more complex than usual, since you must use a boxed definition:

#mul_by_2*N : !{case n : Nat} -> Nat
| succ => succ(succ(mul_by_2(n)))
| zero => zero
halt: zero

#main : !Nat
  ($mul_by_2*)(0n2)





This is explained in more details on the Boxes and Recursion sections of the Tutorial.




Polymorphism

Polymorphic datatypes allow us to create multiple instances of the same datatype with different contained types.

import Base@0

T Pair {A : Type, B : Type}
| pair {x : A, y : B}

main : Num
  let a = pair(~Bool, ~Num, true, 7)

  case/Pair a
  | pair => a.y
  : Num





The {A : Type, B : Type} after the datatype declares two polymorphic variables, A and B, allowing us to create different types of pair, such as a pair of Bool, or a pair of Num, without needing to duplicate the definition of Pair. Each polymorphic variable adds an implicit, erased argument to each constructor, so, instead of pair(true, 7), you need to write pair(~Bool, ~Num, true, 7). In a future, this verbosity will be prevented with implicit arguments. For now, you can amend it with a let:

import Base@0

T Pair {A : Type, B : Type}
| pair {x : A, y : B}

main : [:Pair(Bool, Bool), Pair(Bool, Bool)]
  let pairbb = pair(~Bool, ~Bool)

  let a = pairbb(true, false)
  let b = pairbb(false, true)
  [a, b]





By combining polymorphism with recursive types, we can create the popular List type (already defined on the base libs above):

// {- T List {T : Type}
// | cons {head : T, tail : List(T)}
// | nil

// Returns all but the first element
// tail : {~T : Type, case list : List(T)} -> List(T)
// | cons => list.tail
// | nil  => nil(~T) -}

main : List(Num)
  tail(~Num, cons(~Num, 1, cons(~Num, 2, cons(~Num, 3, nil(~Num)))))





Since List is so common, there is a built-in syntax-sugar for it, the dollar sign:

main : List(Num)
  tail(~Num, Num$[1, 2, 3])








Indices

Indices are like polymorphic variables, except that, rather than constant types, they are computed values that can depend on each constructor’s arguments. That gives us a lot of type-level power, and is one of the reasons Formality is a great proof language. For example:

T IsEven (x : Num)
| make_even {half : Num} (half .*. 2)





This datatype has one index, x, of type Num. Its constructor, is_even, has one field, half : Num. When you write make_even(3), the number 3 is multiplied by two and moved to the type-level, resulting in a value of type IsEven(6). So, for example:

even_0 : IsEven(0)
  make_even(0)

even_2 : IsEven(2)
  make_even(1)

even_4 : IsEven(4)
  make_even(2)

even_6 : IsEven(6)
  make_even(3)





Notice that is impossible to create a value with type IsEven(3), because that would require a half : Num such that half .*. 2 .==. 3, and there is no such number. Because of that, indexed datatypes have many applications. For example, suppose that you want to write a div2 function that divides a number by two, but you don’t want it to be called with odd numbers. You could do this:

div2 : {x : Num, ~x_is_even : IsEven(x)} -> Num
  x ./. 2

main : Num
  div2(10, ~make_even(5))





Here, the second argument “proves” that the first is even, making it impossible to call div2 with an odd input. Since x_is_even is erased from the runtime, this gives us a static, zero-cost guarantee. Alternatively, we could have used ~x_is_even : [k : Num ~ x .==. k .*. 2], but indexed datatypes are more handy in general. For example, you could easily write one for 3 consecutive numbers:

T Consecutive (a : Num, b : Num, c : Num)
| consecutive {init : Num} (init, init .+. 1, init .+. 2)





So, for example, Consecutive(a, b, c) means a, b, c are 3 consecutive numbers such as 7, 8, 9. You could also write a type for sorted lists:

import Base@0

T SortedList {A : Type} (xs : List(Num))
| scons {
  add  : Num,
  head : Num,
  tail : List(Num),
  prof : SortedList(Num, cons(~Num, head, tail))
} (cons(~Num, head .+. add, cons(~Num, head, tail)))
| snil
   (nil(~Num))





Here, SortedList(xs) would mean that xs is a list of Nums with elements in descending order, such as [7, 5, 5, 3, 1]. In order words, a function that returned [xs : List(Num) ~ SortedList(xs)] could only be written if xs was, indeed, sorted. And so on. Indexed datatypes, in a way, give us a powerful language of specifications which we can use to statically reason about our datatypes and algorithms.

For a more complex example of indices, here is a Vector, which is like a List, except that its type stores its own length:

// From Vector@0
// A vector is a list with a statically known length
T Vector {A : Type} (len : Nat)
| vcons {~len : Nat, head : A, tail : Vector(A, len)} (succ(len))
| vnil                                                (zero)





Every time we call vcons, the len index on the type of the vector increases by one, allowing us to track its length statically:

main : Vector(String, 0n3)
   vcons(~String, ~0n2, "ichi",
   vcons(~String, ~0n1, "ni",
   vcons(~String, ~0n0, "san",
   vnil(~String))))





This has many applications. For example, we can create a type-safe vhead that returns the first element of a non-empty vector:

// {-From Vector@0 -}

// {-A type-safe "head" that returns the first element of a non-empty vector
// On the `vcons` case, return the vector's head
// On the `vnil` case, prove it is unreachable, since `xs.len > 0` -}
vhead: {~T : Type, ~n : -Nat, xs : Vector(T, %succ(+n))} -> T
  case/Vector xs
  + refl(~Nat, ~%succ(+n)) as e : Equal(Nat, xs.len, %succ(+n))
  | vcons => xs.head
  | vnil  => absurd(zero_isnt_succ(~n, e), ~T) 
  : T





Notice that constructor indices are accessible on the left side of an equality note. That gives us an e : zero is succ(n) (since zero is the length of vnil, and succ(n) is the length of xs). Since 0 != 1, this branch is unreachable, so we can fill it with an absurd.




Self-Encodings

Interestingly, none of the features above are part of Formality’s type theory. Instead, they are lightweight syntax-sugars that elaborate to plain-old lambdas. To be specific, a datatype is encoded as is own inductive hypothesis, with “Self Types”. For example, the Bool datatype desugars to:

Bool : Type
  ${self}
  { ~P    : {x : Bool} -> Type
  , true  : P(true)
  , false : P(false)
  } -> P(self)

true : Bool
  new(~Bool){~P, true, false} true

false : Bool
  new(~Bool){~P, true, false} false

case_of : {b : Bool, ~P : {x : Bool} -> Type, t : P(true), f : P(false)} -> P(b)
  (%b)(~P, t, f)





Here, ${self} ..., new(~T) val and %b are the type, introduction, and elimination of Self Types, respectively. You can see how any datatype is encoded under the hoods by asking fm to evaluate its type, as in, fm Data.Bool@0/Bool -W (inside the fm_modules directory). The -W flag asks Formality to not evaluate fully since Bool is recursive. While you probably won’t need to deal with self-encodings yourself, knowing how they work is valuable, since it allows you to express types not covered by the built-in syntax.

TODO: write a brief explanation of how Self encodings work (although I think it should be self-explanatory from this example!).







          

      

      

    

  

    
      
          
            
  
Datatypes

Most traditional functional languages of the Haskell/ML family feature algebraic datatypes or ADTs. Those are essentially the bread-and-butter of functional programming. From Lists to Trees to Monads, every data structure in those languages is an ADT. Formality-Core does not include native ADTs; instead, it provides the building blocks for the user to assemble their own datatypes. This is important when it comes to interaction net compilation because the same data structure could be represented in different ways, with different runtime characteristics, each one having benefits and drawbacks. Being explicit about them allows the programmer to have control over the runtime behavior of their programs.


Scott-Encoding

The Scott-Encoding represents ADTs as a series of lambdas. It doesn’t involve boxes and behaves almost identically to traditional ADTs. The advantage of Scott-Encoded datatypes is that they have O(1) elimination, i.e., pattern-matching is a constant-time operation. The disadvantage is that, since Formality-Core is terminating, you can’t iterate/recurse over them directly; instead, you need to use Church-Encoded helpers. The best way to explain a λ-encoding is by example and comparison, so, here are bools, pairs, and trees, with their Haskell-equivalents:

// ::::::::::
// :: Bool ::
// ::::::::::

// data Bool = True | False
// bool_match x true false = case x of { True -> true; False -> false }

def True       : {true false} true
def False      : {true false} false
def bool_match : {x true false} (x true false)

// ::::::::::
// :: Pair ::
// ::::::::::

// data Pair a b = Pair a b
// pair_match x pair = case x of { Pair a b -> pair a b }

def Pair       : {a b pair} (pair a b)
def pair_match : {x pair} (x {a b} (pair a b))

// ::::::::::
// :: Tree ::
// ::::::::::

// data Tree a = Node (Tree a) (Tree a) | Leaf a
// tree_match x node leaf = case x of { Node a b -> node a b; Leaf val -> leaf val }

def Node       : {a b node leaf} (node a b)
def Leaf       : {val node leaf} (leaf val)
def tree_match : {x node leaf} (x {a b} (node a b) {val} (leaf val))

// ::::::::::::::
// :: Examples ::
// ::::::::::::::

def main:

  let bool_ex =
    let bool       = True
    let true_case  = "Bool is true" 
    let false_case = "Bool is false"
    (bool_match bool true_case false_case)

  let pair_ex =
    let pair      = (Pair 4 5)
    let pair_case = {a b} ["Pair sum is:", |a + b|] 
    (pair_match pair pair_case)

  let tree_ex =
    let tree      = (Node (Leaf 42) (Leaf 64))
    let node_case = {a b} "Tree is a Node"
    let leaf_case = {val} "Tree is a Leaf"
    (tree_match (Node (Leaf 42) (Leaf 64)) node_case leaf_case)

  [bool_ex,
  [pair_ex,
    tree_ex]]





Thanks to the Scott-Encoding, any Haskell or Agda program that doesn’t involve duplications or recursion can be translated directly to Formality-Core without changes. For iteration and recursion, we need the Church-Encoding.




Church-Encoding

The Church-Encoding is like a materialization of a Haskell fold. It involves boxes, and is used for bounded iteration and recursion. Without Church-Encodings, Formality-Core would be a boring language where all its programs could do is move/permutate stuff around and stop in linear time. They’re only required for variable-size (i.e., recursive) datatypes, since, on the fixed-size case, Church and Scott are identical. Once again, they’re better explained through examples:

// ::::::::::
// :: List ::
// ::::::::::

// data List a = Cons a (List a) | Nil
// list_fold x cons nil = case x of { Cons x xs -> cons x (list_fold xs succ zero); Nil -> nil }

def Nil: {cons nil}
  dup cons = cons
  dup nil  = nil
  # nil

def Cons: {x xs cons nil}
  dup cons = cons
  dup nil  = nil
  dup x    = x
  dup xs   = (xs #cons #nil)
  # (cons x xs)

def fold_list: {x cons nil}
  (x cons nil)

// ::::::::::
// :: Tree ::
// ::::::::::

// data Tree a = Node (Tree a) (Tree a) | Leaf a
// tree_fold x node leaf = case x of { Node a b -> node (tree_fold a node leaf) (tree_fold b node leaf); Leaf val -> leaf val }

def Node: {a b node leaf}
  dup node = node
  dup leaf = leaf
  dup a    = (a #node #leaf)
  dup b    = (b #node #leaf)
  # (node a b)

def Leaf: {val node leaf}
  dup node = node
  dup leaf = leaf
  dup val  = val
  # (leaf val)

def fold_tree: {x node leaf}
  (x node leaf)

// ::::::::::::::
// :: Examples ::
// ::::::::::::::

def main:

  let list_ex =
    let list      = (Cons #1 (Cons #2 (Cons #3 (Cons #4 Nil))))
    let fold_cons = # {x xs} |x + xs|
    let fold_nil  = # 0
    (fold_list list fold_cons fold_nil)

  let tree_ex =
    let tree      = (Node (Node (Leaf #1) (Leaf #2)) (Node (Leaf #3) (Leaf #4)))
    let fold_node = # {a b} |a + b|
    let fold_leaf = # {val} val
    (fold_tree tree fold_node fold_leaf)

  [list_ex, tree_ex]





Notice how, unlike on the Scott-Encoded versions, here we are able to fold over the structures, summing all their contained values. Moreover, due to the use of boxes, elements contained by a Church-Encoded structure must always be one layer above than them, which, by consequence, means that results of folds are one layer above too.







          

      

      

    

  

    
      
          
            
  
Formality Net

Formality terms are compiled to a memory-efficient interaction net system.
Interaction nets are just graphs where nodes have labeled ports, one being the
main one, plus a list of “rewrite rules” that are activated whenever two nodes
are connected by their main ports. Our system includes 6 types of nodes, ERA,
CON, OP1, OP2, ITE, NUM.

[image: ../_images/fm-net-node-types.png]


	CON has 3 ports and an integer label. It is used to represent lambdas,
applications, boxes (implicitly) and duplications. Since FM-Core is based on
EAL, there is no book-keeping machinery to keep track of Bruijn indices, just
CON is enough for beta-reduction.


	ERA has 1 port and is used to free empty memory, which happens when a
function that doesn’t use its bound variable is applied to an argument.


	NUM has 1 port and stores an integer and is used to represent native
numbers.


	OP1 has 2 ports and stores one integer and an operation id. OP2 has 3
ports and an operation id. They are used for numeric operations such as
addition and multiplication.


	ITE has 3 ports and an integer label. It is used for if-then-else, and is
required to enable number-based branching.




Note that the position of the port matters. The port on top is called the main
port. The first port counter-clockwise to the main port (i.e., to the left on
this drawing) is the aux0 port, and the first port clockwise to the main port
(i.e., to the right on this drawing) is the aux1 port.


Rewrite rules

In order to perform computations, FM-Net has a set of rewrite rules that are
triggered whenever two nodes are connected by their main ports. This is an
extensive list of those rules:

[image: ../_images/fm-net-rewrite-rules.png]

Note that, while there are many rules (since we need to know what to do on each
combination of a node), most of those have the same “shape” (such as OP2-OP2,
ITE-ITE), so they can reuse the same code. There are only 5 actually relevant
rules:


Erasure

When an ERA or a NUM node collides with anything, it “destroys” the other
node, and propagates itself to destroy all nodes connected to it.




Substitution

When two CON nodes of equal label collide, and also on the OP2-OP2 /
ITE-ITE cases, both nodes are destroyed, and their neighbors are connected.
That’s the rule that performs beta-reduction because it allows connecting the
body of a lambda (which is represented with CON) to the argument of an
application (which is, too, represented with CON). Note that on the OP2-OP2
and ITE-ITE cases, that’s just a default rule that doesn’t matter, since those
cases can’t happen on valid FM-Core programs.




Duplication

When different nodes collide, they “pass through” each other, duplicating
themselves in the process. This allows, for example, CON nodes with a label
>1 to be used to perform deep copies of any term, with dup x = val; .... It
can copy lambdas and applications because they are represented with CON nodes
with a label 0, pairs and pair-accessors, because they are represented with
CON nodes with a label 1, and ITE, OP1, OP2, because they are
different nodes.

It also allows duplications to duplicate terms that are partially duplicated
(i.e., which must duplicate, say, a λ-bound variable), as long as the CON
labels are different, otherwise, the CON nodes would instead fall in the
substitution case, destroying each other and connecting neighbors, which isn’t
correct. That’s why FMC’s box system is necessary: to prevent concurrent
duplication processes to interfere with each other by ensuring that, whenever
you duplicate a term with dup x = val; ..., all the duplication CON nodes of
val will have a labels higher than the one used by that dup.




If-Then-Else

When an ITE node collides with a NUM node, it becomes a CON node with one
of its ports connected to an ERA node. That’s because then/else branches are
actually stored in a pair, and this allows you to select either the fst or the
snd value of that pair and discard the other branch.




Num-Operation

When OP2 collides with a NUM, it becomes an OP1 node and stores the number
inside it; i.e., the binary operation becomes a unary operation with NUM
partially applied. When that OP1 collides with another NUM, then it performs
the binary operation on both operands, and return a new NUM with the result.
Those rules allow us to add, multiply, divide and so on native numbers.






Compiling FM-Core to FM-Net

The process of compiling FM-Core to FM-Net can be defined by the following
function k_b(net):

[image: ../_images/fm-net-compilation.png]

This function recursively walks through a term, creating nodes and “temporary
variables” (x_b) in the process. It also keeps track of the number of boxes it
passed through, b. For example, on the lambda ({x}f) case, the procedure
creates a CON node with a label 0, creates a “temporary variable” x_b on
the aux0 port, recurses towards the body of the function, f on the aux1
port, and then returns the main port (because there is a black ball on it).
Notice that there isn’t a case for VAR. That’s what those “temporary
variables” are for. On the VAR case, two things can happen:


	If the corresponding “temporary variable” x_b was never used, simply return
a pointer to it.


	If the corresponding “temporary variable” x_b was used, create a “CON” node
with a label 2 + b, connect its main port to the old location of x_b, its
aux0 to the port x_b pointed to, and return a pointer to its aux1 port.




This process allows us to create as many CON nodes as needed to duplicate
dup-bound variables, and labels those nodes with the layer of that dup (plus
2, since labels 0 and 1 are used for lambdas/applications and
pairs/projections). Note that this process is capable of duplicating λ-bound
variables, but this isn’t safe in practice, and won’t happen in well-typed
inputs.




Implementation

In our implementation, we use a buffer of 32-bit unsigned integers to represent
nodes, as follows:


	CON: represented by 4 consecutive uints. The first 3 represent the main,
aux0 and aux1 ports. The last one represents the node type (3 bits),
whether its ports are pointers or unboxed numbers (3 bits), and the label (26
bits).


	OP1: represented by 4 consecutive uints. The first and third represent
the main and aux0 ports. The second represents the stored number. The
last one represents the node type (2 bits), whether its ports are pointers
or unboxed numbers (3 bits, 1 unused), and the operation (26 bits).


	OP2: represented by 4 consecutive uints. The first 3 represent the main,
aux0 and aux1 ports. The last one represents the node type (3 bits),
whether its ports are pointers or unboxed numbers (3 bits), and the operation
(26 bits).


	ITE: represented by 4 consecutive uints. The first 3 represent the main,
aux0 and aux1 ports. The last one represents the node type (3 bits),
whether its ports are pointers or unboxed numbers (3 bits), and the label
(26 bits).


	ERA: is stored inside other nodes and do not use any extra space. An ERA
node is represented by a pointer port which points to itself. That’s because
ERA’s rewrite rules coincide with what we’d get if we allowed ports to point
to themselves.


	NUM: is stored inside other nodes and do not use any extra space. A NUM
node is represented by a numeric port. In order to know if a port is a number
or a pointer, each node reserves 3 bits of its last uint to store that
information.





Rewrites

TODO: explain how rewrite works, how link_ports is used, and why
unlink_ports is necessary to avoid invalid states.




Strict evaluation

The strict evaluation algorithm is very simple. First, we must keep a set of
redexes, i.e., nodes connected by their main ports. In order to do that,
whenever we link two main ports, we must add the address of the smallest nodes
to that set. We then perform a loop to rewrite all redexes. This will give us a
new set of redexes, which must then be reduced again, over and over, until there
are no redexes left. This is the pseudocode:

while (len(net.redexes) > 0):
  for redex in net.redexes:
    net.rewrite(redex)





The strict reduction is interesting because it doesn’t require graph walking nor
garbage collection passes, and because the inner for-loop can be performed in
parallel. That is, every redex in net.redexes can be rewritten at the same
time.

In order to do that, though, one must be cautious with intersection areas. For
example, in the graph below, B-C and D-E are redexes. If we reduce them in
parallel, both threads will attempt to read/write from C’s and D’s aux0 and
aux1 ports, potentially causing synchronization errors.

[image: ../_images/sk_problem_2x.png]

This can be avoided through locks, or by performing rewrites in two steps. On
the first step, each thread reads/writes the ports of its own active pair as
usual, except that, when it would need to connect a neighbor, it instead turns
its own node into a “redirector” which points to where the neighbor was supposed
to point. For example, substitution and duplication would be performed as
follows:

[image: ../_images/sk_local_rewrites_2x.png]

Notice that P, Q, R and S (neighbor ports) weren’t touched: they keep
pointing to the same ports, but now those ports point to where they should point
to. Then, a second parallel step is performed. This time, we spawn a thread for
each neighbor port and walk through the graph until we find a non-redirector
node. We then point that neighbor to it. Here is a full example:

[image: ../_images/sk_local_rewrites_ex_2x.png]

Notice, for example, the port C of the node A. It is on the neighborhoods of
a redex (B-C), but isn’t a redex itself. On the first step, two threads
rewrite the nodes B-C and D-E, turning them into redirectors, and without
touching that port. On the second step, a thread starts from port C of node
A, towards port B of node B (a redirector), towards port C of node D
(a redirector), towards port B of node F. Since that isn’t a redirector,
the thread will make C point to B. The same is done for each neighbor port
(in parallel), completing the parallel reduction.




Lazy evaluation

The lazy evaluation algorithm is very different from the strict one. It works by
traversing the graph, exploring it to find redexes that are “visible” on the
normal form of the term, skipping unnecessary branches. It is interesting
because it allows avoiding wasting work; for example, ({a b}b (F 42) 7) would
quickly evaluate to 7, no matter how long (F 42) takes to compute. In
exchange, it is “less parallel” than the strict algorithm (we can’t reduce all
redexes since we don’t know if they’re necessary), and it requires global
garbage collection (since erasure nodes are ignored).

To skip unnecessary branches, we must walk through the graph from  port to port,
using a strategy very similar to the denotational semantics of symmetric
interaction combinators. First, we start walking from the root port to its
target port. Then, until we get back to the root, do as follows:


	If we’re walking from an aux port towards an aux port of a node, add the aux
we’re coming from to a stack, and move towards the main port of that node.


	If we’re walking from a main port to an auxiliary port, then we just found a
node that is part of the normal form of the graph! If we’re performing a
weak-head normal form reduction, stop. Otherwise, start walking towards each
auxiliary port (either recursively, or in parallel).


	If we’re walking towards root, halt.




This is a rough pseudocode:

def reduce_lazy(net, start):
  back = []
  prev = start
  next = net.enter(prev)

  while not(net.is_root(next)):
    if slot_of(prev) == 0 and slot_of(next) == 0:
      net.rewrite(prev, next)
    elif slot_of(next) == 0:
      for aux_n from 0 til net.aux_ports_of(next):
        net.reduce_lazy(Pointer(node_of(next), aux_n))
    else:
      back.push(prev)
      prev = Pointer(node_of(next), 0)
      next = net.enter(prev)





While the lazy algorithm is inherently sequential, there is still an opportunity
to explore parallelism whenever we find a node that is part of the normal form
of the graph (i.e., case 2). In that case, we can spawn a thread to walk
towards each auxiliary port in parallel; i.e., the for-loop of the pseudocode
can be executed in parallel like the one on the strict version. This would allow
the algorithm to have many threads walking through the graph at the same time.
Again, caution must be taken to avoid conflicts.









          

      

      

    

  

    
      
          
            
  
Self-Types

(TODO)





          

      

      

    

  

    
      
          
            
  
Dependent Types

Obs: Work in progress

Cover things like:


	Returning a different type based on the function’s input

iff : {x : Bool, ~A : Type, t : A, f : A} -> A
  (use(x))(~{x}A, t, f)

make_unit_or_word : {x : Bool} -> iff(x, ~Type, Unit, Word)
  case/Bool x
  | true  => unit
  | false => 42
  : iff(x, ~Type, Unit, Word)







	Manipulating equalities

b_is_true : {a : Bool, b : Bool, b_is_a : b == a, a_is_true : a == true} -> b == true
  b_is_a :: rewrite a in b == a with a_is_true







	Specifying precise algorithm (for work outsourcing)

// {- "I want a function that receives a bool and returns a different bool" -}
Specification : Type
  {a : Bool} -> [b : Bool, ~Not(a == b)]

// {- "This is a valid implementation of the Specification" -}
not : Specification 
  case/Bool a
  | true  => [false , ~true_isnt_false]
  | false => [true  , ~{e} true_isnt_false(sym(~e))]
  : [b : Bool, ~Not(self == b)]







	Exploiting impossible cases to improve function’s interface

// {- If we know that a Maybe isn't none, we can extract its contents -}
extract : {~A : Type, x : Maybe(A), not_none : Not(x == none(~A))} -> A
  case/Maybe x
  | just => val
  | none => absurd(not_none(refl(~none(~A))), ~A)
  : A









etc.





          

      

      

    

  

    
      
          
            
  
Recursion


Boxed definitions

Since bounded recursive functions are so common, Formality has built-in syntax for them, relying on “boxed definitions”. To make a boxed definition, prepend # to its name. That has two effects. First, the whole definition is lifted to level 1. Second, it allows you to use boxed definitions inside ($...)’s: the parser will automatically unbox them for you. For example, instead of this:

foo : !Word
  #40

bar : !Word
  #2

main : !Word
  dup foo = foo
  dup bar = bar
  # foo .+. bar





We could write:

#foo : !Word
  40

#bar : !Word
  2

#main : !Word
  ($foo) .+. ($bar)





Both programs are the same, except the later is shorter.




Recursion

Formality allows you to turn a boxed definition into a recursive function by appending *N to its name (where N is a new variable, which will track how many times the function was called), and adding a “halt-case” with a halt: term on the last line (where term is an expression that will be returned if the function hits its call limit). So, for example, a factorial function could be written as:

#fact*N : ! {i : Word} -> Word
  if i .==. 0:
    1
  else:
    i .*. fact(i .-. 1)
halt: 0





This is not much different from the usual fact definition, except we explicitly set the “halt-case” to be 0 on the last line. That means that, if the function “runs out of gas”, it will stop and return 0 instead. As a shortcut, if your “halt-case” is simply one of the function’s argument, you can write the * on it instead, as in, fact*N ! {*i : Word} -> Word. To call it, you must set an explicit max call limit with *N:

main : !Word
  dup f = fact*100
  # f(12)





Or, with boxed definitions:

#main : !Word
  ($fact*100)(12)





The f*N syntax configures the call limit of a recursive function. Here, we used 100. Note this is actually just a shortcut for a function application: we could have written fact(*100) instead. We could also have omitted the number as in ($fact*)(x), which would default to 2^256-1. This limit is so absurdly large that, for all practical purposes, our functions are no less powerful than the ones found in other languages. After all, 2^256-1 is so large that no real computer could reach this amount of calls anyway. In fact, the entire observable universe has less particles than that!




Structural Recursion

When it comes to inductive proofs, the need for a halt-case on Formality’s recursive function syntax can be limited to deal with. For example, you can’t easily prove that add(a,b) = add(b,a), since that is not actually true when the add function hits its call limit! The more traditional structural recursion that other languages feature is more flexible in that sense. There are interesting workarounds this problem, such as writing add in a way that consumes both sizes simultaneously. A general solution, though, is still missing from the language, as we’re debating the best way to do it.

We have some great candidates, though. As an example, we could use Bound, as detailed on this commit [https://github.com/moonad/Formality-Base/commit/b777d806c6fa37f2ce306fbe87b3ed267152b90c]. It allows us to prove that our arguments are decreasing in size, essentially emulating Coq’s structural recursion. This is really cool, as it allows us to avoid writing the (provably unreachable) halt-case! But it still requires a lot of manual boilerplate to do correctly. If you don’t want that, we suggest you to avoid writing complex inductive proofs in Formality for now. In the future, we’ll probably add syntax sugars for structural recursion (or something equivalent), making those proofs much less cumbersome.

Obs: Work in progress

Cover things like:


	Simple recursive functions and boxed definitions

// {- From Nat@0 -}
#double*N : !{case halt n : Nat} -> Nat
| succ => succ(succ(double(n.pred)))
| zero => zero

#double.example : !Nat
  ($double*)(succ(zero))







	Polymorphic recursive functions with level-0 parameters

// {- From List@0 -}
#map*N : {~A : Type, ~B : Type, f : !A -> B} -> ! {case list : List(A)} -> List(B)
| cons => cons(~B, f(list.head), map(list.tail))
| nil  => nil(~B)
halt: nil(~B)

#map.example : !List(Word)
  ($map*(~Word, ~Word, #{x} x + 1))(Word$[1,2,3,4])







	Indexed recursive functions using N




... vector stuff, fin stuff, etc...





etc.







          

      

      

    

  

    
      
          
            
  
Linearity

Formality’s approach to termination is what makes it different from other proof languages like Agda, Idris and Coq. Instead of having native datatypes, structural recursion and so on, we go deeper and change the underlying logic of the system from intuitionist to elementary affine. This is responsible for all the claimed benefits of Formality: optimal reductions, no garbage-collection, massive parallelism, elegant inductive types and so on. But it comes with a huge tradeoff: our lambdas are affine, i.e., bound variables can’t be used more than once. This limits what we can do in multiple ways. For example, we can’t write a function that ands a Bool with itself:

import Base@0

self_and : {x : Bool} -> Bool
  and(x, x)

main : Bool
  self_and(true)





If we try to check the type using fm -t <file_name>/main, we get an error:

[ERROR]
Use of affine variable `x` more than once in proof-relevant position.
- When checking x
- With the following context:
- x : Bool





There are multiple ways to avoid this situation.


Reuse the variable on multiple branches

If the “duplicated” variable is used in different branches, as in:

import Base@0

// {- Negates `b` if `a` is true -}
not_if : {a : Bool, b : Bool} -> Bool
  case/Bool a
  | true  => not(b)
  | false => b
  : Bool

main : Bool
  not_if(true, true)





Then, we can avoid copying b with a clever trick: return a lambda on each branch. Like this:

import Base@0

// {- Negates `b` if `a` is true -}
not_if : {a : Bool, b : Bool} -> Bool
  (case/Bool a
  | true  => {b} not(b)
  | false => {b} b
  : Bool -> Bool)(b)

main : Bool
  not_if(true, true)





Notice that instead of using b directly, the case/Bool expression returns, in each case, a different lambda, which is then applied to a single b. This prevents using it more than once, and is allowed. This technique is extremelly important for Formality development.




Use case’d arguments

While the trick above is powerful, it increases code complexity. Fortunately, if you use case’d arguments instead of case/T expressions, Formality will automatically do it for you. For example, this works:

import Base@0

// {- Negates `b` if `a` is true -}
not_if : {case a : Bool, b : Bool} -> Bool
| true  => not(b)
| false => b

main : Bool
  not_if(true, true)





And is much less verbose than the solution above. In practice, this features allows you to use a variable once per branch of the function, instead of once per function.




Make an explicit copy

For Words in particular, there is a native cpy operation that copies it as many times as desired:

square : {x : Word} -> Word
  cpy x = x
  x .*. x

main : Word
  square(7)





When a Word is an argument of a top-level function, then you don’t even need to add cpy. Formality does it for you. I.e., this works:

square : {x : Word} -> Word
  x .*. x

main : Word
  square(7)





For other types, you can write an auxiliary copy function:

import Base@0

// {- An explicit copying function -}
copy_bool : {b : Bool} -> [:Bool, Bool]
  case/Bool b
  | true  => [true, true]
  | false => [false, false]
  : [:Bool, Bool]

and_itself : {b : Bool} -> Bool
  get [b0, b1] = copy_bool(b) // performs an explicit copy
  and(b0, b1)

main : Bool 
  and_itself(true)





This is also a very important technique. So, in short, when you need to use a variable more than once, this is what you should do:


	Is it a Word? If so, just cpy it.


	Is the usage in different branches? Then manually return lambdas (or use case’d arguments).


	Otherwise, copy the structure with an explicit copy function.







Use boxes

Formality has another primitive for deep-copying values, boxes. When dealing with data, though, you almost never want to use boxes to perform copies, due to the stratification condition, which essentially segregates the language in levels, blocks communication from higher to lower levels. Regardless, they can still be useful sometimes. See, for example, how map is defined for lists:

#map*n : {~A : Type, ~B : Type, f : !A -> B} -> ! {case list : List(A)} -> List(B)
| cons => cons(~B, f(list.head), map(list.tail))
| nil  => nil(~B)
halt: nil(~B)





Here, f is duplicated on level 0, allowing it to be used multiple times on level 1. The tradeoff is that relative to programs on level 1, f must be seen as static. So, if a user input arrives on level 0, for example, it can’t affect the shape of f.

Boxes really shine when implementing control flow like loops and recursion. This will be explained in more detail in the next section.







          

      

      

    

  

    
      
          
            
  
Boxes

As explained previously, Formality includes primitives for performing deep copies of boxed terms:




	syntax
	effect





	#t
	Puts term t inside a box



	!T
	The type of a boxed term



	dup x = t; u
	Unboxes t and copies it as x inside u




The dup primitive is the one responsible for copying and is extremely important as it performs those copies lazily, in a way that allows optimal sharing of sub-expressions. It is what makes Formality a great closure evaluator. But, in order to use it properly, you must understand how it is limited: the stratification condition.


The Stratification Condition

The primitives above, without restriction, would be dangerous. For example, this:

main
  let f = {x}
    dup x = x
    x(#x)
  f(#f)





Would loop forever, which should never happen in a terminating language. To solve that, Formality relies on the stratification condition. In short, it enforces the following invariant:


The level of a term can never change during the program evaluation.




Where the level of a term is the number of boxes “wrapping” it.


Counting the level of a term

To understand the restriction above, you must be able to count the level of a term. Let’s do it on the following example:

["a", #"b", "c", #["d", #"e"], ##"f"]






	The string "a" isn’t wrapped by any box. It is on level 0.


	The string "b" is wrapped by one box. It is on level 1.


	The string "c" isn’t wrapped by any box. It is on level 0.


	The string "d" is wrapped by one box. It is on level 1.


	The string "e" is wrapped by two boxes (one indirect). It is on level 2.


	The string "f" is wrapped by two boxes. It is on level 2.




The type of the program above is:

[:String, :!String, :String, :![:String, !String], !!String]








Stratification examples

This condition is imposed globally, forbidding certain programs. For example:

box : {x : Word} -> !Word 
  # x





This isn’t allowed because, otherwise, we would be able to increase the level of a word. Similarly, this:

main : [:Word, Word]
  dup x = #42
  [x, x]





Isn’t allowed too, because 42 would jump from level 1 to level 0 during runtime. But this:

main : ![:Word, Word]
  dup x = #42
  # [x, x]





Is fine, because 42 remains on level 1 after being copied. And this:

main : [:!Word, !Word]
  dup x = #42
  [#x, #x]





Is fine too, for the same reason.






Applications


Copying “static” data

In general of this, boxes aren’t very useful for copying data. That’s because information can only flow from lower to higher levels. So, for example, if some piece of data is generated on level 2, you can copy it on level 3, but you can’t use it again on level 2. Generally, your program’s logic should stay on the highest level, with the lower levels being used to copy static data and generate bounded-depth recursive functions. In fact, Formality’s syntax sugars and standard libraries are designed to be used with two levels only: the level 0, where recursive functions are created and static data is duplicated, and the level 1, where everything is used. So, for example, Data.List/map function on Base uses level 0 to make multiple copies of f, which are used on level 1:

#map*n : {~A : Type, ~B : Type, f : !A -> B} -> ! {case list : List(A)} -> List(B)
| cons => cons(~B, f(list.head), map(list.tail))
| nil  => nil(~B)
halt: nil(~B)








Implementing loops/recursion

While Formality has a built-in syntax for recursion, it can be insightful to understand how it is implemented under the hoods. Mind the following program:

ten_times : {~T : Type, f : !{x : T} -> T, x : !T} -> !T
  dup f = f
  dup x = x
  # f(f(f(f(f(f(f(f(f(f(x))))))))))

main : !Word
  ten_times(~Word, #{x} x .+. 2, #0)





Here, we define a function, ten_times, which takes a function, f, creates 10 copies of it, and applies to an argument, x. As a result, we’re able to repeat the + 2 operation 10 times, adding 20 to 0. This same technique can be used to implement bounded recursion. For example, here:

ten_times : {~T : Type, f : !{x : T} -> T, x : !T} -> !T
  dup f = f
  dup x = x
  # f(f(f(f(f(f(f(f(f(f(x))))))))))

fact : !{n : Word} -> Word
  let call = {rec, i}
    cpy i = i
    if i .==. 0:
      1
    else:
      (i .*. rec(i .-. 1))
  let halt = {i}
    0
  ten_times(~{x : Word} -> Word, #call, #halt)

main : !Word
  dup fact = fact
  # fact(6)





We “emulate” a recursive function by using ten_times to “build” the recursion tree of “fact” up to 10 layers deep. As such, it only works for inputs up to 10; after that, it hits the “halt” case and returns 0. The good thing about this way of doing recursion is that we’re not limited to recurse on structurally smaller arguments. The bad thing is that it is a little bit verbose, requiring an explicit bound, and a halting case for when the function “runs out of gas”. Moreover, since we used ten_times to make the function, it comes inside a box, on level 1. In other words, it is impossible to use it on level 0! Instead, we must use the level 0 to unbox it (with a dup), and then use it on level 1. As usual, you could simplify it with a boxed definition:

#main : !Word
  ($fact)(6)





Formality’s recursion syntax builds a similar program, except:


	Instead of a hard-coded max call limit, it is configurable *N.


	Instead of simple repetition, it uses Nat induction, allowing you to use the call count, N, in types.




So, for example, when you write:

#fact*N : ! {i : Word} -> Word
  if i .==. 0:
    1
  else:
    i .*. fact(i .-. 1)
halt: 0





It gives you a fact : {N : Ind} -> !{n : Word} -> Word, instead of a fact : !{n : Word} -> Word. You can call it inside a boxed definition with ($fact*MAX_CALLS)(x).









          

      

      

    

  

    
      
          
            
  
Theorem Proving

TODO: review and update with correct error messages (after the Hole update). Also talk about log(x).

TODO: this article is outdated since there isn’t an equality primitive anymore. It must be updated to use Data.Equal instead.


Simple proofs

Let’s prove a theorem about the boolean not:

import Base@0

main : {b : Bool} -> Equal(Bool, %not(not(b)), %b)
  ?a





Evaluate it using fm -t <file>/main and the type checker complains:

Found hole: 'a'.
- With goal... Equal(Bool, not(not(b)), b)
- Couldn't solve it.
- With context:
- b : Bool

Unsolved holes.





Let’s pattern-match on b.

import Base@0

main : {b : Bool} -> Equal(Bool, %not(not(b)), %b)
  case/Bool b
  | true  => ?t
  | false => ?f
  : Equal(Bool, %not(not(b)), %b)





Not helpful:

Found hole: 't'.
- With goal... Equal(Bool, not(not(true)), true)
- Couldn't solve it.
- With context:
- b : Bool

Found hole: 'f'.
- With goal... Equal(Bool, not(not(false)), false)
- Couldn't solve it.
- With context:
- b : Bool

Unsolved holes.





If we reduce both sides, we get the same expression: {true, false} => true. In this case, we can use refl:

import Base@0

main : {b : Bool} -> Equal(Bool, %not(not(b)), %b)
  case/Bool b
  | true  => refl(~Bool, ~%true)
  | false => ?f
  : Equal(Bool, %not(not(b)), %b)





Progress, compiler now complains about the false branch:

Found hole: 'f'.
- With goal... Equal(Bool, not(not(false)), false)
- Couldn't solve it.
- With context:
- b : Bool

Unsolved holes.





We can do the same:

import Base@0

main : {b : Bool} -> Equal(Bool, %not(not(b)), %b)
  case/Bool b
  | true  => refl(~Bool, ~%true)
  | false => refl(~Bool, ~%false)
  : Equal(Bool, %not(not(b)), %b)





No type error. Our proof is complete! If we used case’d args, the proof becomes just:

import Base@0

main : {case b : Bool} -> Equal(Bool, %not(not(b)), %b)
| true  => refl(~Bool, ~%true)
| false => refl(~Bool, ~%false)








Inductive proofs

Let’s prove a similar theorem, but for negation on arbitrary-length bit-strings instead of plain booleans:

import Base@0

// {-  From Bits@0
// T Bits
// | b0 {pred : Bits}
// | b1 {pred : Bits}
// | be-}

#bnot*n : !{case halt bits : Bits} -> Bits
| b0 => b1(bnot(bits.pred))
| b1 => b0(bnot(bits.pred))
| be => be





Start with the theorem we want to prove:

#main*n : !{bits : Bits} -> Equal(Bits, %($bnot(n))(($bnot(n))(bits)), %bits)
  ?a
  halt: ?b





Remember that halt: is mandatory on recursive definition to provide the base-case. TODO: explain why the -#s on the type.

The type checker complains:

Found hole: 'a'.
- With goal... Equal(Bits, $bnot(step(n))($bnot(step(n))(bits)), bits)
- Couldn't solve it.
- With context:
- n    : Ind
- n    : -Ind
- main : {bits : Bits} -> Equal(Bits, $bnot(n)($bnot(n)(bits)), bits)
- bits : Bits

Found hole: 'b'.
- With goal... Equal(Bits, $bnot(base)($bnot(base)(bits)), bits)
- Couldn't solve it.
- With context:
- n    : Ind
- bits : Bits

Unsolved holes.





Notice that:


	It asks for P(step(n)) instead of P(n).


	We have, on context, main, which gives us P(n).




That’s because the body of a recursive function is actually the step case of inductive proof, so all we need to do is, assuming P(n), prove P(step(n))!

Let’s match against bits:

#main*n : !{bits : Bits} -> Equal(Bits, %($bnot(n))(($bnot(n))(bits)), %bits)
  case/Bits bits
  | b0 => ?b0
  | b1 => ?b1
  | be => ?be
  : Equal(Bits, %($bnot(n))(($bnot(n))(bits)), %bits)
  halt: ?b





Now the complaint becomes:

Type mismatch.
- Found type... Hole
- Instead of... bnot(step(n), bnot(step(n), b0(pred))) == b0(pred)
- When checking ?
- On expression {bits} => ?
- With the following context:
- n    : Ind
- n    : Ind
- main : {bits : Bits} -> bnot(n, bnot(n, bits)) == bits
- bits : Bits
- pred : Bits





This is better because now it expects b0(pred) instead of just bits . This allows the left-side of the equation to be reduced to:

b0(bnot(n, bnot(n, pred))) == b0(pred)





This is perfect because we can use the inductive hypothesis to get this same equation, without the b0s. As in, we need to go…

from :    bnot(n, bnot(n, bs))  == b0(bs)
to   : b0(bnot(n, bnot(n, bs))) ==    bs





All we need is to add b0 on both sides. We can do it with cong, from the base libraries (Base@0):

#main*n : !{bits : Bits} -> <bnot(n)>(<bnot(n)>(bits)) == bits
  case/Bits bits
  | b0 => cong(~Bits, ~Bits, ~(<bnot(n)>)((<bnot(n)>)(pred)), ~pred, ~b0, ~main(pred))
  | b1 => ?
  | be => ?
  : <bnot(step(n))>(<bnot(step(n))>(self)) == self
  halt: ?





TODO: should we have a built-in syntax to simplify cong?

Now the checker complains about the b1 case:

Type mismatch.
- Found type... Hole
- Instead of... bnot(step(n), bnot(step(n), b1(pred))) == b1(pred)
- When checking ?
- On expression {pred} => ?
- With the following context:
- n    : Ind
- n    : Ind
- main : {bits : Bits} -> bnot(n, bnot(n, bits)) == bits
- bits : Bits
- pred : Bits





We can easily complete this proof now:

#main*n : !{bits : Bits} -> <bnot(n)>(<bnot(n)>(bits)) == bits
  case/Bits bits
  | b0 => cong(~Bits, ~Bits, ~<bnot(n)>(<bnot(n)>(pred)), ~pred, ~b0, ~main(pred))
  | b1 => cong(~Bits, ~Bits, ~<bnot(n)>(<bnot(n)>(pred)), ~pred, ~b1, ~main(pred))
  | be => refl(~be)
  : <bnot(step(n))>(<bnot(step(n))>(self)) == self
  halt: refl(~bits)





As usual, it could be simplified with case’d arguments:

#main*n : !{case bits : Bits} -> <bnot(n)>(<bnot(n)>(bits)) == bits
| b0 => cong(~Bits, ~Bits, ~<bnot(n)>(<bnot(n)>(bits.pred)), ~bits.pred, ~b0, ~main(bits.pred))
| b1 => cong(~Bits, ~Bits, ~<bnot(n)>(<bnot(n)>(bits.pred)), ~bits.pred, ~b1, ~main(bits.pred))
| be => refl(~be)
halt: refl(~bits)





Note that the big difference here, with relation to Agda/Coq proofs, is that, in their cases, since recursive functions are defined by structural recursion, inductive proofs are also defined by recursion on the structure. For example, if we wanted to prove this theorem in Agda, we’d just match the bit-string, prove the base case by reflexivity, and prove the recursive case by calling main recursively on pred.

In Formality, it looks like the proof is the same, but there is a subtle, yet important, difference: under the hoods, we’re not actually recursing on the Bits structure. Instead, we’re folding over n : Ind, a datatype capturing the inductive hypothesis on natural numbers. As such, in order to prove that bnot(n, bnot(n, bits)) == bits hold for any n, we first must prove that it is true for n == 0, i.e., when bits has a maximum recursion depth of 0 (i.e., is “out-of-gas”), which is true by reflexivity since the function returns bits itself. We then prove that assuming this is true for a maximum recursion depth of n, then it is also true for bits(step(n)). This is the step case, which coincides with Agda’s proof.

Proving that kind of inductive theorem on Formality is a little more verbose than in Agda, since you have to wrap the whole proof inside Ind, and always tell the compiler what to do when the function “runs out of gas”. In exchange, since termination is guaranteed by EAL, there is no “structural recursion” checker, so you’re allowed to be more flexible in your recursive definitions.







          

      

      

    

  

    
      
          
            
  
Formality-Core Tutorial

This tutorial aims to teach how to effectively develop Formality-Core code,
assuming experience in functional programming languages, in special Haskell.
Formality-Core is a minimal language based on elementary affine logic, making it
compatible with optimal reductions. It can be seen as the GHC Core to the
upcoming Formality language. Its minimalism, the lack of a type system, and its
unusual boxed duplication system make it a very bare-bones language that isn’t
easy to work with directly, demanding that the programmer learns some delicate
techniques to be productive.


1. Core features

Before proceeding, you should have Formality
installed, and be familiar with its core features
and syntax. If you’re not yet, please read the entire Language
section of this documentation.




2. Simple datatypes

Algebraic Datatypes (ADTs) are the building bricks of functional programming
languages like Haskell, where all programs are just functions operating on ADTs.
Programming in Formality isn’t fundamentally different. As such, the best way to
learn it is by learning how to translate Haskell code.

Let’s start with a simple type: booleans. In Haskell, they can be defined with
the following declaration:

{-# LANGUAGE NoImplicitPrelude #-}

data Bool
  = True
  | False





This puts 2 constructors, True and False in scope. In Formality, there is no
data syntax, instead we use T:

T Bool
| True
| False





Let’s attempt to translate a simple function, not:

not :: Bool -> Bool
not True  = False
not False = True





The first thing we must do is get rid of Haskell’s equational notation (which
Formality-Core doesn’t have) in favor of lambdas and case-ofs, like this:

not :: Bool -> Bool
not = \ a -> case a of {
  True  -> False;
  False -> True;
}





For the sake of clarity, it is also recommended that each case is given a name
with a let, as follows:

not :: Bool -> Bool
not = \ a ->
  let case_True  = False in
  let case_False = True in
  (case a of { True -> case_True; False -> case_False })





Once a Haskell program is in this shape, translating it to Formality is
straigthforward: we just have to adjust the syntax.

not : {x : Bool} -> Bool
  case<Bool> x
  | True  => False
  | False => True
  : Bool





Run the program below with fm <file_name>.main.

T Bool
| True
| False

not : {x : Bool} -> Bool
  case<Bool> x
  | True  => False
  | False => True
  : Bool

main : Bool
  not(False)





As an exercise, implement bool_to_nat, which returns 1 or 0.




3. Efficient branching

Let’s now translate the following and function:

-- Exhaustive patterns for the sake of demonstration
and :: Bool -> Bool -> Bool
and True  True  = True
and False True  = False
and True  False = False
and False False = False





Converting the syntax to Formality:

and : {a : Bool, b: Bool} -> Bool
  case<Bool> a
  | True  => b
  | False => False
  : Bool

and2 : {|a : Bool, |b: Bool} -> Bool
| True  | True  = True
        | False = False
| False | True  = False
        | False = False





Here is another example:

let swap    = 0 // Change to 1 to swap
let big_arr = [ 0, [ 1, [ 2, [ 3,[ 4,[ 5,[ 6,[ 7, 8]]]]]]]]
let big_str = "I'm a big string with a lot of stuff!"
(if swap
  then: {big_arr big_str} [big_str, big_arr]
  else: {big_arr big_str} [big_arr, big_str]
  big_arr big_str)





This snippet creates an array and a string and then makes a pair of both with
either the string first or the array first, depending on the value of swap.
Despite using big_str and big_arr in both branches, those values were never
copied thanks to the technique above.

Run the program below with fmc main.

def True: {True False}
  True

def False: {True False}
  False

def and: {a b}
  let case_a_True = {b}
    let case_b_True  = True
    let case_b_False = False
    (b case_b_True case_b_False)
  let case_a_False = {b}
    let case_b_True  = False
    let case_b_False = False
    (b case_b_True case_b_False)
  (a case_a_True case_a_False b)

def main: 
  let bool = (and True False)

  // Prints the value of Bool
  let case_True  = "I'm true!"
  let case_False = "I'm false!"
  (bool case_True case_False)





As an exercise, implement or.




4. Datatypes with fields

In Haskell, pairs can be defined as:

{-# LANGUAGE NoImplicitPrelude #-}

data Pair a b
  = NewPair a b





This puts 1 constructor, NewPair, in scope. This is the corresponding
Formality-Core definitions:

NewPair {a, b} {NewPair}
  NewPair(a, b)





Notice that those are mostly similar to True and False, except now there are
two fields, {a b}, involved. Let’s write the first accessor function in the
Formality-ready form:

fst :: Pair a b -> a
fst = \ pair -> 
  let case_NewPair = \ a b -> a in
  case pair of { NewPair a b -> case_NewPair a b }





Notice that, here, each field of the datatype became a lambda on the
case_NewPair expression. This is the Formality translation:

def fst: {pair}
  let case_NewPair = {a b} a
  (pair case_NewPair)





Run the program below with fmc main.

def NewPair: {a b} {NewPair}
  (NewPair a b)

def get_first: {pair}
  let case_NewPair = {a b} a
  (pair case_NewPair)

def main: 
  let pair = (NewPair 1 2)

  // Prints the first element of `pair`
  (get_first pair)





As an exercise, implement pair_swap.




5. Non-recursive datatypes

Here are 2 more Haskell datatypes:

data Maybe a
  = Nothing
  | Just a

data Either a b
  = Left a
  | Right b





This puts 4 constructors, Nothing, Just, Left, Right in scope. Those are
the corresponding Formality-Core definitions:

def Nothing: {Nothing Just}
  Nothing

def Just: {a} {Nothing Just}
  (Just a)

def Left: {a} {Left Right}
  (Left a)

def Right: {b} {Left Right}
  (Right b)





From this, you should be able to grasp the general pattern:

def Ctor_0: {ctor_0_field_0 ctor_0_field_1 ...} {Ctor_0 Ctor_1 ...}
  (Ctor_0 ctor_0_field_0 ctor_0_field_1 ...)

def Ctor_1: {ctor_1_field_0 ctor_1_field_1 ...} {Ctor_0 Ctor_1 ...}
  (Ctor_1 ctor_1_field_0 ctor_1_field_1 ...)

...





Let’s now implement a from_just :: Maybe a -> Either String a function that
either extracts the value of a Maybe, or returns an error:

def from_just: {error_msg maybe_val}
  let case_nothing = (Left error_msg)
  let case_just    = {val} (Right val)
  (maybe_val case_nothing case_just)





By this point, you should be able to understand this. The general pattern for
matching is:

let case_Ctor_0 = {ctor_0_field_0 ctor_0_field_1 ...} result_on_case_Ctor0
let case_Ctor_1 = {ctor_1_field_0 ctor_1_field_1 ...} result_on_case_Ctor1
...





Run the program below with fmc main.

def Nothing: {Nothing Just}
  Nothing

def Just: {a} {Nothing Just}
  (Just a)

def Left: {a} {Left Right}
  (Left a)

def Right: {b} {Left Right}
  (Right b)

def from_just: {error_msg maybe_val}
  let case_nothing = (Left error_msg)
  let case_just    = {val} (Right val)
  (maybe_val case_nothing case_just)

def main:
  let maybe_a = Nothing
  let maybe_b = (Just 3)
  (from_just "I'm not a number." maybe_a)





As an exercise, implement to_maybe :: Either a b -> Maybe b.




6. Unboxed copying

Remember that Formality-Core functions can only use its bound variable once.
Because of that, it is hard to make duplicates of a value. For example, this
isn’t possible:

// square : Nat -> Nat
def square: {n}
  |n * n|





Here, we learned how to avoid this problem when the copy is needed in a
different branch, but what if we really need the value twice, like on the case
above? Fortunately, as explained on the
wiki [https://github.com/moonad/Formality/wiki/Dups-and-Boxes], Formality-Core
includes an explicit duplication system that allows us to write this:

// square : !Nat -> !Nat
def square: {n}
  dup n = n
  # |n * n|





But this kind of definition has an important limitation: it can only affect a
number in a layer below it! In general, a term on layer N can’t read any
information from a term on layer N+1. Because of that, you often want all
data of your program to live in a single layer and avoid boxes as much as
possible. But then, how do we copy values? On the square case, what you want
to do is to use the cpy primitive, which allows you to copy a number as many
times as you want:

// square : Nat -> Nat
def square: {n}
  cpy n = n
  |n * n|





For user-defined algebraic datatype, you must write an explicit copy function,
which performs a pattern-match and explicitly returns copies of the same value.
For example:

def copy_bool: {b}
  let case_true  = [True, True]
  let case_false = [False, False]
  (b case_true case_false)

def main:
  let bool = True
  get [bool_cpy_0, bool_cpy_1] = (copy_bool bool)
  ...





This is an annoying complication that could and should be automated on the
Formality language. When dealing with Formality-Core, writing explicit copy
functions is one of the programmer’s job. As an exercise, write a
copy_bool_pair function that copies a pair of bools (i.e., (copy_bool_pair [True, False]) == [[True,False], [True,False]]).




7. Recursive datatypes

In Haskell, natural numbers and linked lists can be defined as:

{-# LANGUAGE NoImplicitPrelude #-}

data Nat
  = Succ Nat
  | Zero

data List a
  = Cons a (List a)
  | Nil





This puts 4 constructors, Succ, Zero, Cons, Nil in scope. Those are the
corresponding Formality-Core definitions:

def Succ: {n} {Succ Zero}
  (Succ n)

def Zero: {Succ Zero}
  Zero

def Cons: {x xs} {Cons Nil}
  (Cons x xs)

def Nil: {Cons Nil}
  Nil





There is no surprise here: the definitions are identical to non-recursive
datatypes. What changes, though, is that we can’t use those datatypes as
expected. This, for example, won’t work:

def length: {list}
  let case_cons = {x xs} (Succ (length xs))
  let case_nil  = Nil
  (list case_cons case_nil)





The problem is that, being a terminating language, recursion isn’t allowed.
Recursive datatypes without recursive functions are of limited use. What now?

That’s when boxes become useful. Remember that I told you to avoid using boxes
to copy data? That’s because the real use of boxes is to capture loops, folds,
and recursion through the so-called “Church-Encodings”. This is best explained
through examples. To recursive over a list of length up to 4, this is how you do
it:

def Cons: {x xs} {Cons Nil}
  (Cons x xs)

def Nil: {Cons Nil}
  Nil

def rec4: {call stop}
  dup call = call
  dup stop = stop
  # (call (call (call (call stop))))

def length:
  let call = {rec list}
    let case_cons = {x xs} |1 + (rec xs)|
    let case_nil  = 0
    (list case_cons case_nil)
  let stop = 0
  dup length = (rec4 #call #stop)
  # {list}
    (length list)

def main:
  dup length = length
  # let list = (Cons 7 (Cons 7 Nil))
    ["Length of the list is:", (length list)]





Let’s take a moment to understand what is going on here because this is very
important and unusual. First, we create rec4, which is the “Church-Encoded”
natural number 4, which allows us to execute a recursive function up to 4 calls.
Then, we implement length using it. There, we define call, which is exactly
what the usual Haskell length function would look like, except that it
receives an extra argument, rec, to refer to itself. Then, we define stop,
which is the “default” value in the case the recursion hits its call limit
(here, 4). Then we create the length function by applying rec4 to #call
and #stop. That function will only work up to 4 calls. Since we used dups on
layer 0, this length function will only be available on layer 1. That’s
why there is a # before {list}: our entire function is written on layer 1.
On this case, we just receive a list and apply length to it.

Crazy, right? This is, by far, the most confusing aspect of Formality, so, don’t
get afraid if you don’t get it at first. In general, though, you can expect
Formality-Core programs to follow roughly the form of the length function.
First, we “configure” the recursive calls and their “call limits” on layer 0,
and then we use them on layer 1. From my experience, under normal circumstances,
Formality-Core programs should use no more than two layers. Take a look at
Kaelin [https://github.com/moonad/Formality/blob/master/stdlib/kaelin.fmc], a
small game written in Formality-Core: recursive functions like write,
update, vec2_range are “configured” on layer 0, while the game logic and its
data live on layer 1. (Note: actually, right now, it is on layer 2, but
disregard as this will be changed soon.)

Note: since recN is so common, in order to avoid writing it for each n,
there is quick syntax-sugar to generate it. First, define rec as:

def rec: {n call stop}
  dup callN = (n call)
  dup stop  = stop
  # (callN stop)





Then, you can write (rec ~N) for any N.




8. Fusing loops

The ~ syntax generates a ultra-compact Church Nats. For example, ~256
becomes:

{s}
  (dup s0 = s
  (dup s1 = #{x} (s0 (s0 x))
  (dup s2 = #{x} (s1 (s1 x))
  (dup s3 = #{x} (s2 (s2 x))
  (dup s4 = #{x} (s3 (s3 x))
  (dup s5 = #{x} (s4 (s4 x))
  (dup s6 = #{x} (s5 (s5 x))
  (dup s7 = #{x} (s6 (s6 x))
  (dup s8 = #{x} (s7 (s7 x))
    #{x} (s8 x))))))))))





Instead of the full form (with 256 applications of s). This compact form is
useful for getting asymptotical speed-ups over traditional functional languages.
For example, the program below:

def True: {True False}
  True

def False: {True False}
  False

def not: {a True False}
  let case_True  = False
  let case_False = True
  (a case_True case_False)

def main: 
  let num      = ~1000000000000
  dup num_nots = (num #not)
  # (num_nots True "num is even" "num is odd")





Applies not one trillion times to the boolean True and returns instantly if
running on interaction nets (fmc -l main). As an experiment, you can try
changing the number to anything else and it will always output whether num is
even or odd, even a computer shouldn’t be able to perform 1 trillion function
calls that quickly. That’s one of the cool aspects of Formality and boils down
to its ability to perform fusion (like Haskell’s foldr/build technique) at
runtime, merging compositions of not and essentially performing a loop of N
iterations in O(log(N)) graph-rewrites. I write more about this effect on
this
article [https://medium.com/@maiavictor/solving-the-mystery-behind-abstract-algorithms-magical-optimizations-144225164b07],
and implement a quick, elegant exp_mod on this
Gist [https://gist.github.com/MaiaVictor/e556062185c5863d814980123e03630f].



Uff, that’s a lot of information already! What now? Help me improving this
guide [https://github.com/moonad/docs.formality-lang.org/issues]

